
Maple-assisted derivation of recurrence for A181252

There are  possible rows.  We enumerate them so that row  consists of the binary digits of 
 (with leading 0's included as needed). 

Consider the  transition matrix  with entries if the rows of a  sub-array could be 
in configurations  and configuration , and 0 otherwise.  The following Maple code computes it.   

for i from 1 to 512 do Configs[i]:= convert(2^9+i-1,base,2)[1..9]
od:
Compatible:= proc(i,j)
 if `and`(seq(evalb(Configs[i][k] + Configs[i][k+1] + Configs[j]
[k]+Configs[j][k+1] < 4), k=1..8)) then 1 else 0 fi
end proc:
T:= Matrix(512,512,Compatible):

Thus for   where  is a column vector with all entries 1.  
v:= Vector(512,1):

To check, here are the first few entries of our sequence.  For future use, we pre-compute more  than 
we need.

TV[0]:= v:
for nn from 1 to 150 do TV[nn]:= T . TV[nn-1] od:
seq(v^%T . TV[n],n=0..23);

The recurrence shows up as a linear dependence among  We gather these as columns of a matrix , 
and stop when it has less than full column rank.

L:= v:
for nn from 1 do
  L:= <L|TV[nn]>;

  if LinearAlgebra:-Rank(L) < nn+1 then printf("Success at n=
%d\n",nn); break fi;
od:

Success at n=48

The recurrence can then be found from the null space of the matrix .
P:= LinearAlgebra:-NullSpace(L)[1]:

This is the recurrence:
recurrence:= sort(a(n)=solve(add(P[i]*a(n+i-49),i=1..49),a(n)),
[seq(a(n-i),i=0..49)]);

